



# Documentation Technique – KPIs & Mesures DAX (Power BI)

Bah Cheikh | Responsbele Data | Équipe Data

## Contexte & conventions

Qu'est-ce qu'un RDV ? Rendez-vous **généré après une conversation** (appel connecté) entre un commercial et un prospect, puis **programmé** à une date/heure. Le jour J :

- **RDV fait** : le prospect se présente, l'échange a lieu.
- **No Show** : le prospect ne vient pas.
- **RDV annulé** : le RDV est annulé avant l'heure prévue.

**Objectif RDVs** : nombre de RDVs qu'un commercial doit réaliser sur une période donnée.

**Sources** : rdvs, hubspot-data, notion-data, dim-clients, image-commerciaux. **Syntaxe**

**DAX** : doubler l'apostrophe dans les noms de colonnes (ex. 1"appel).

## Niveau 1 — Mesures métiers directes

### RDVs (volumétrie de base)

#### Total RDVs

*Définition* : total des RDVs programmés, toutes issues confondues. *Utilité* : volumétrie d'opportunités créées. *Lecture* : suivre par commercial/source/période.

```
Total RDVs =  
COUNT ( rdvs[Statut] ) - [RDVs à faire]
```

## RDVs fait

*Définition* : RDVs tenus (réalisés). *Utilité* : mesure de résultat opérationnel réel. *Lecture* : à rapprocher du taux de no show/annulation.

```
RDVs fait =  
CALCULATE ( COUNT ( rdvs[Statut] ), rdvs[Statut] = "RDV fait" )
```

## RDVs confirmé

*Définition* : RDVs confirmés avant la date. *Utilité* : indicateur de qualité de planification. *Lecture* : un

```
RDVs Confirmé =  
CALCULATE ( COUNT ( rdvs[Statut] ), rdvs[Statut] = "RDV confirmé" )
```

## RDVs annulé

*Définition* : RDVs annulés avant l'heure. *Utilité* : pertes post-programmation. *Lecture* : analyser par source et délai entre prise et tenue.

```
RDVs annulé =  
CALCULATE ( COUNTROWS ( rdvs ), rdvs[Statut] = "RDV annulé" )
```

## No Show

*Définition* : RDVs où le prospect ne s'est pas présenté. *Utilité* : pertes à la dernière étape. *Lecture* : agir via rappels, SMS, créneaux courts.

```
No_Show =  
CALCULATE ( COUNTROWS ( rdvs ), rdvs[Statut] = "No Show" )
```

## Appels HubSpot (alimentation du pipe)

### Total appels HubSpot

*Définition* : volume d'appels initiés. *Utilité* : effort commercial brut. *Lecture* : à normaliser par temps disponible/portefeuille.

```
Total appels HubSpot =  
COUNT ( 'hubspot-data'[id] )
```

## Total appels connecté

*Définition* : appels aboutis à une conversation. *Utilité* : qualité de contact. *Lecture* : base de conversion vers RDVs.

```
Total appels connecté =  
CALCULATE (  
    COUNT ( 'hubspot-data'[Résultat de l'appel] ),  
    'hubspot-data'[Résultat de l'appel] = "Connecté"  
)
```

## Taux de décrochage HubSpot

*Définition* : part d'appels connectés. *Utilité* : efficacité de joignabilité. *Lecture* : faible taux → revoir base, horaires, scripts.

```
Taux_decrochages_hubspot_ =  
DIVIDE ( [Total appels connecté], [Total appels HubSpot] )
```

## Notion (agrégats de pilotage)

### RDVs pris

*Définition* : RDVs programmés durant la période. *Utilité* : output du funnel “conversation → RDV”. *Lecture* : à comparer à l'objectif.

```
To_RDVs_Pris = SUM ( 'notion-data'[RDVs] )
```

### Total Appels (Notion)

*Définition* : effort déclaré. *Utilité* : cohérence avec HubSpot. *Lecture* : écart → process de saisie.

```
Total_Appels = SUM ( 'notion-data'[Appels] )
```

### Total Conversations (Notion)

*Définition* : échanges aboutis. *Utilité* : pont vers conversion RDV. *Lecture* : trop bas → qualification fichiers.

```
Total_Conversations = SUM ( 'notion-data'[Conversations] )
```

## Objectif RDVs

*Définition* : cible de RDVs de la période. *Utilité* : pilotage managérial. *Lecture* : sert au taux de réalisation.

```
Objectif_RDVs = SUM ( 'notion-data'[Objectif_RDVs] )
```

## Objectif RDV Client (dimension)

*Définition* : objectif au niveau client/compte. *Utilité* : vues segmentées. *Lecture* : agrégation par portefeuille.

```
Objectif_RDV_Client = CALCULATE ( SUM ( clients[Objectif_RDVs] ) )
```

## Chiffre d'affaire

*Définition* : CA généré par sessions. *Utilité* : lien RDV → valeur. *Lecture* : formaté pour affichage.

```
Chiffre d'affaire =  
FORMAT ( CALCULATE ( SUM ( 'notion-data'[CA_GH_Session] ) ), "#,##0 " )
```

## Délai génération → tenue

### Délai (texte pour affichage)

*Définition* : jours moyens entre génération et RDV fait (format texte). *Utilité* : lisible sur tuile. *Lecture* : viser court (synchro agenda).

```
Délai (Génér -> RDV fait) =  
FORMAT (  
    AVERAGEX (  
        FILTER ( rdvs, rdvs[Statut] = "RDV fait" ),  
        DATEDIFF ( rdvs[Date_generation], rdvs[DateTime_RDV], DAY )  
    ),  
    "0"  
) & " jour"
```

## Délai (numérique, jours)

*Définition* : même KPI en valeur numérique pour calculs (scores). *Utilité* : agrégation, bornage. *Lecture* : exploitable dans une formule.

```
Délai (Génér -> RDV) [jours] =  
AVERAGEX (  
  FILTER ( rdvs, rdvs[Statut] = "RDV fait" ),  
  DATEDIFF ( rdvs[Date_generation], rdvs[DateTime_RDV], DAY )  
)
```

## Niveau 2 — Mesures complémentaires (dérivées)

### RDVs non fait (volume)

*Définition* : annulé + no show. *Utilité* : pertes post-programmation. *Lecture* : action : confirmations, rappels, requalif.

```
RDVs non fait = [RDVs annulé] + [No_Show]
```

### Total RDVs annulé (auxiliaire)

*Définition* : auxiliaire pour taux. *Utilité* : base de % annulé. *Lecture* : segmentation par source.

```
Total RDVs annulé =  
CALCULATE ( COUNT ( rdvs[Statut] ), rdvs[Statut] = "RDV annulé" )
```

### Taux de RDVs fait

*Définition* : part des RDVs réalisés. *Utilité* : efficacité finale. *Lecture* : cible haute.

```
Taux de RDVs fait =  
DIVIDE ( [RDVs fait], [Total RDVs], 0 )
```

### Taux de RDV annulé

*Définition* : part des RDVs annulés. *Utilité* : friction avant J. *Lecture* : baisser via reconfirmations.

```
Taux de RDV annulé =  
DIVIDE ( [Total RDVs annulé], [Total RDVs] )
```

## Taux de No Show

*Définition* : part des RDVs no show. *Utilité* : risque dernier mètre. *Lecture* : baisser via SMS/rappels.

```
Taux de No Show =  
DIVIDE ( [No_Show], [Total RDVs] )
```

## Taux de RDVs non fait

*Définition* : annulé + no show en %. *Utilité* : KPI synthèse de perte. *Lecture* : doit décroître dans le temps.

```
Taux de RDVs non fait =  
[Taux de RDV annulé] + [Taux de No Show]
```

## Taux de conversion RDVs

*Définition* : RDVs pris / conversations. *Utilité* : qualité du pitch/qualification. *Lecture* : comparer par script/canal.

```
Taux_Conversion_RDVs =  
DIVIDE ( SUM('notion-data'[RDVs]), SUM('notion-data'[Conversations]), 0 )
```

## Taux de décrochage (Notion)

*Définition* : conversations / appels. *Utilité* : joignabilité. *Lecture* : horaires, bases, cadence.

```
Taux de décrochage =  
DIVIDE ( SUM('notion-data'[Conversations]), SUM('notion-data'[Appels]), 0 )
```

## Taux de Réalisation Objectifs

*Définition* : RDVs pris / objectif. *Utilité* : pilotage managers. *Lecture* : >100% = dépassement.

```
Taux Réalisation Objectifs =  
DIVIDE ( [To_RDVs_Pris], [Objectif_RDVs] )
```

## Durée moyenne des appels connectés

*Définition* : durée moyenne des appels “connectés”. *Utilité* : proxy de qualité d’échange. *Lecture* : trop court = pitch faible ; trop long = dispersion.

```

Durée_Moyenne_ =
VAR totalSecondes =
    CALCULATE (
        AVERAGE ( 'hubspot-data'[Durée_Secondes] ),
        'hubspot-data'[Résultat de l'appel] = "Connecté"
    )
VAR heures = QUOTIENT ( totalSecondes, 3600 )
VAR resteMinutes = MOD ( totalSecondes, 3600 )
VAR minutes = QUOTIENT ( resteMinutes, 60 )
VAR secondes = MOD ( resteMinutes, 60 )
RETURN
FORMAT(heures, "00") & ":" & FORMAT(minutes, "00") & ":" & FORMAT(secondes, "00")

```

### Moyenne différence entre appels

*Définition* : espacement moyen entre appels vs durée moyenne. *Utilité* : cadence de relance.  
*Lecture* : trop grand → perte de chaleur.

```

Moyenne_Différence_entre_Appels =
VAR MoyenneDiff = AVERAGE ( 'hubspot-data'[Différence entre les appels (secondes)] )
VAR DureeMoyenne = [Durée_Moyenne_]
VAR Delta = MoyenneDiff - DureeMoyenne
VAR Heures = INT ( Delta / 3600 )
VAR Minutes = INT ( MOD ( Delta, 3600 ) / 60 )
VAR Secondes = MOD ( Delta, 60 )
RETURN
FORMAT(Heures, "00") & ":" & FORMAT(Minutes, "00") & ":" & FORMAT(Secondes, "00")

```

## Niveau 3 — Mesures d'aide UI

### NB jours sélectionné

*Utilité* : contrôler l'état des filtres.

```

NB jours sélectionné =
COUNTROWS ( VALUES ( 'notion-data'[Date & Heure Session].[Jour] ) )

```

### NB de Commercial sélectionnée

*Utilité* : idem, côté commerciaux.

```

NB de Commercial sélectionnée =
COUNTROWS ( VALUES ( 'image-commerciaux (2)'[Commercial] ) )

```

## Text du carré

*Utilité* : message d'aide si pas de jour sélectionné.

```
Text du carré =  
IF ( [NB jours sélectionné] = 1, "", "Veuillez sélectionner un jour sur la graphe"  
    " )
```

## Titre carré commercial

*Utilité* : incitation à choisir un commercial.

```
titre caré commercial =  
IF ( [NB de Commercial sélectionnée] = 1, "", " Choisir un commercial  afficher  
    ses statistiques" )
```

## Rang jour

*Utilité* : classement des jours par réalisation d'objectifs.

```
Rang jour =  
RANKX (   
    ALL ( 'notion-data'[Date & Heure Session].[Jour] ),  
    CALCULATE ( DIVIDE ( [To_RDVs_Pris], [Objectif_RDVs] ) ),  
    ,  
    DESC,  
    Skip  
    )
```



## Variation Taux Réalisation

*Utilité* : libellé “amélioration/baisse/stable”.

```
Variation Taux Réalisation =  
VAR DateActuelle =  
    MAXX ( ALLSELECTED('notion-data'), INT('notion-data'[Date & Heure Session]) )  
VAR DatePrecedente =  
    CALCULATE (  
        MAXX (  
            FILTER ( ALL('notion-data'),  
                INT('notion-data'[Date & Heure Session]) < DateActuelle ),  
            INT('notion-data'[Date & Heure Session])  
        )  
    )  
VAR TauxActuel =  
    CALCULATE ( [Taux Réalisation Objectifs],  
        FILTER ( ALL('notion-data'),  
            INT('notion-data'[Date & Heure Session]) = DateActuelle ) )  
VAR TauxPrecedent =  
    CALCULATE ( [Taux Réalisation Objectifs],  
        FILTER ( ALL('notion-data'),  
            INT('notion-data'[Date & Heure Session]) = DatePrecedente ) )  
VAR Variation =  
    IF ( NOT ISBLANK(TauxActuel) && NOT ISBLANK(TauxPrecedent),  
        TauxActuel - TauxPrecedent, BLANK() )  
RETURN  
IF ( ISBLANK(Variation), "-",  
    IF ( Variation > 0,  
        UNICHAR(9650) & " Amélioration de" & FORMAT(Variation, "+0.00%"),  
        IF ( Variation < 0,  
            UNICHAR(9660) & " Baisse de " & FORMAT(Variation, "0.00%"),  
            FORMAT(Variation, "0.00%") ) ) )
```

## Variation Color

*Utilité* : code couleur simple pour tuiles.

```
Variation Color =
VAR DateActuelle =
  MAXX ( ALLSELECTED('notion-data'), INT('notion-data'[Date & Heure Session]) )
VAR DatePrecedente =
  CALCULATE (
    MAXX (
      FILTER ( ALL('notion-data'),
        INT('notion-data'[Date & Heure Session]) < DateActuelle ),
      INT('notion-data'[Date & Heure Session]) ) )
VAR TauxActuel =
  CALCULATE ( [Taux Réalisation Objectifs],
    FILTER ( ALL('notion-data'),
      INT('notion-data'[Date & Heure Session]) = DateActuelle ) )
VAR TauxPrecedent =
  CALCULATE ( [Taux Réalisation Objectifs],
    FILTER ( ALL('notion-data'),
      INT('notion-data'[Date & Heure Session]) = DatePrecedente ) )
VAR Variation = TauxActuel - TauxPrecedent
RETURN IF ( ISBLANK(Variation), "neutral", IF ( Variation > 0, "good", "bad" ) )
```

## Couleur Performance RDV semaine à semaine

*Utilité* : feedback visuel J vs J-7.

```
Couleur Performance RDV semaine à semaine =
VAR DateActuelle = INT ( MAX ( 'notion-data'[Date & Heure Session] ) )
VAR DatePrecedente =
  MAXX (
    FILTER ( ALL('notion-data'),
      INT('notion-data'[Date & Heure Session]) <= DateActuelle - 7 ),
    INT('notion-data'[Date & Heure Session]) )
VAR TauxActuel =
  CALCULATE ( [Taux Réalisation Objectifs],
    FILTER ( ALL('notion-data'),
      INT('notion-data'[Date & Heure Session]) = DateActuelle ) )
VAR TauxPrecedent =
  CALCULATE ( [Taux Réalisation Objectifs],
    FILTER ( ALL('notion-data'),
      INT('notion-data'[Date & Heure Session]) = DatePrecedente ) )
VAR Variation =
  IF ( NOT ISBLANK(TauxActuel) && NOT ISBLANK(TauxPrecedent),
    TauxActuel - TauxPrecedent, BLANK() )
RETURN
IF ( ISBLANK(Variation), "Gray",
  IF ( Variation > 0, "Blue",
    IF ( Variation < 0, "Red", "Black" ) ) )
```

## NbSessions

*Utilité* : charge de sessions par jour/tranche.

```
NbSessions =  
COUNTROWS (  
  SUMMARIZE (  
    ADDCOLUMNS (  
      'notion-data',  
      "DateJour", DATEVALUE('notion-data'[Date & Heure Session]),  
      "TrancheHoraire",  
      IF ( HOUR('notion-data'[Date & Heure Session]) < 13, "Matin", "Après-midi  
" )  
    ),  
    [DateJour], [TrancheHoraire]  
  )  
)
```

## nbclients

*Utilité* : taille de base clients.

```
nbclients = COUNTROWS ( 'dim-clients' )
```

## Niveau 4 — Scores

### Score\_RDVs\_NonFaits

*But* : pénaliser simultanément le **taux** et le **volume** de non-faits. *Lecture* : plus le score est élevé, mieux c'est (peu de pertes).

```
Score_RDVs_NonFaits =  
VAR Taux = [Taux de RDVs non fait]  
VAR Volume = [RDVs non fait]  
VAR MinVolume = 1  
VAR MaxVolume = 8  
VAR VolumeNormInverse =  
  IF ( Volume <= MinVolume, 1,  
      IF ( Volume >= MaxVolume, 0,  
          DIVIDE ( MaxVolume - Volume, MaxVolume - MinVolume ) ) )  
VAR Score = (1 - Taux) * (0.5 + 0.5 * VolumeNormInverse) * 100  
RETURN ROUND ( Score, 1 )
```

## Score\_RDVs\_généré

*But* : récompenser volume de RDVs pris, tenue et conversion. *Lecture* : pondérations ajustables selon priorités.

```
Score_RDVs_généré =  
VAR Taux_RDVs = [Taux de RDVs fait]  
VAR Taux_Conversion = [Taux_Conversion_RDVs]  
VAR Volume = [To_RDVs_Pris]  
VAR MinVolume = 8  
VAR MaxVolume = 66  
VAR VolumeNorm =  
  IF ( Volume <= MinVolume, 0,  
      IF ( Volume >= MaxVolume, 1,  
          DIVIDE ( Volume - MinVolume, MaxVolume - MinVolume ) ) )  
VAR WeightVolume = 0.5  
VAR WeightTaux = 0.25  
VAR WeightConversion = 0.25  
VAR Score =  
  ( WeightVolume * VolumeNorm +  
    WeightTaux * Taux_RDVs +  
    WeightConversion * Taux_Conversion ) * 100  
RETURN ROUND ( Score, 1 )
```

## Score\_Delai

*But* : encourager la rapidité (2–21 jours). *Lecture* : 1 = très rapide, 0 = trop long.

```
Score_Delai =  
VAR Delai = [Délai (Génér -> RDV) [jours]]  
VAR Dmax = 21  
VAR Dmin = 2  
VAR DureeCorrigee = MAX ( 0, Dmax - Delai )  
RETURN ROUND ( DureeCorrigee / (Dmax - Dmin), 3 )
```

## Score Global

*But* : synthèse pondérée pour classement. *Lecture* : sert au cadrage coaching/bonus.

```
Score Global =  
[Taux Réalisation Objectifs] * 0.15 +  
[Score_RDVs_généré] * 0.35 +  
[Score_RDVs_NonFaits] * 0.30 +  
[Score_Delai] * 0.20
```

## Rang Score Global

*But* : classement des commerciaux. *Lecture* : comparer dans le temps et par segment.

```
Rang Score Global =  
RANKX (  
  ALL ( 'image-commerciaux (2)' [Commercial] ),  
  [Score Global], , DESC, Skip  
)
```

## Rang (badge texte)

*But* : rendu lisible dans l'UI. *Lecture* : 1er/2e/3e, sinon ne.

```
Rang (badge texte) =  
VAR rang = [Rang Score Global]  
RETURN  
SWITCH ( rang, 1, "1er", 2, "2e", 3, "3e", rang & "e" )
```

## Conseils d'usage

- Aligner **Objectif RDVs** par canal/ancienneté ; ajuster mensuellement.
- Attaquer **No Show/Annulé** : confirmations, rappels J-1, créneaux rapprochés, SMS.
- Réduire le **Délai** : proposer un créneau < 7 jours, relances automatiques.
- Améliorer **Conversion** : travailler le script, la qualification, les raisons de refus.
- Piloter par **Score Global** : cibler 1–2 actions par commercial, mesurer l'impact la semaine suivante.